Коллеги, приветствую.
Есть управляемая форма, у которой основной реквизит это НаборКонстант. Программно добавляю элемент формы, и путь к данным у него хочу указать как "НаборКонстант.МояКонстанта", но платформа ругается на неправильность пути.
Что делаю не так?
---
Спасибо!
Вот, выдержка с ИТС

По умолчанию свойство Использовать всегда установлено для всех реквизитов формы, за исключением случаев:

Константы. Если добавить в конфигурацию новую константу, то при установленном свойстве она будет передаваться во все формы констант, которые были созданы раньше и ничего об этой константе не "знали". А это не только лишний трафик, но и необходимость иметь права на чтение данной константы. Поэтому для констант свойство Использовать всегда по умолчанию не установлено.

Получается программно добавить элемент формы и связать его с константой нельзя.
---
код выглядит таким образом (путь) Форма.НаборКонстант.МояКонстанта ?
а так полный код в студию)))
---
```1c
    Элемент = Элементы.Добавить(
    "ИспользоватьЧтоТо",
    Тип("ПолеФормы"),
    ГруппаИспользованияЧегоТО);
    Элемент.Вид = ВидПоляФормы.ПолеФлажка;
    Элемент.ПутьКДанным = "НаборКонстант.ИспользоватьЧтоТо";
```
---
верно. только в НаборКонстант галочки "использовать всегда", на тех константах которыми будешь пользоваться, тогда будет счастье.
---
Спасибо!
Вот, выдержка с ИТС

По умолчанию свойство Использовать всегда установлено для всех реквизитов формы, за исключением случаев:

Константы. Если добавить в конфигурацию новую константу, то при установленном свойстве она будет передаваться во все формы констант, которые были созданы раньше и ничего об этой константе не "знали". А это не только лишний трафик, но и необходимость иметь права на чтение данной константы. Поэтому для констант свойство Использовать всегда по умолчанию не установлено.

Получается программно добавить элемент формы и связать его с константой нельзя.
---
Получается программно добавить элемент формы и связать его с константой нельзя
Можно через произвольный реквизит, как то так:

```1c
    //Добавление константы на форму
    &НаСервере
    Процедура Команда1НаСервере(ИмяКонстанты)
    //Извлекаем константу
    Константа = Константы[ИмяКонстанты].Получить();
    //Определяем тип
    ТипРеквизита = ТипЗнч(Константа);
    //Проверяем метаданные
    Если Метаданные.НайтиПоТипу(ТипРеквизита) = Неопределено Тогда
    //Простой тип: Дата Строка и т.д.
    СтрокаТип = Строка(ТипРеквизита);
    Иначе
    //Оъекты типа СправочникСсылка. ПеречислениеСсылка. и т.д.
    СтрокаТип = СтрЗаменить(Метаданные.НайтиПоТипу(ТипРеквизита).ПолноеИмя(),".","Ссылка.");
    КонецЕсли;
    //Описываем новый реквизит формы
    РеквизитНовый = Новый РеквизитФормы(ИмяКонстанты, Новый ОписаниеТипов(СтрокаТип) );
    ДобавляемыеРеквизиты = Новый Массив;
    ДобавляемыеРеквизиты.Добавить(РеквизитНовый);
    //Добавляем реквизит в форму
    ЭтаФорма.ИзменитьРеквизиты(ДобавляемыеРеквизиты);
    //Создаем элемент формы для нового реквизита
    НовыйЭлемент = ЭтаФорма.Элементы.Добавить(ИмяКонстанты, Тип("ПолеФормы"));
    НовыйЭлемент.ПутьКДанным = ИмяКонстанты;
    НовыйЭлемент.Вид = ВидПоляФормы.ПолеВвода;
    НовыйЭлемент.Заголовок = Метаданные.Константы[ИмяКонстанты].Синоним;
    //Заполняем элемент
    ЭтаФорма[ИмяКонстанты] = Константа;
    //Обработчик изменения
    НовыйЭлемент.УстановитьДействие("ПриИзменении","ЭлементПриИзменении");
    КонецПроцедуры

    &НаКлиенте
    Процедура Команда1(Команда)
    // Событие на нажатие "чегото"
    // Команда1НаСервере("ДатаТекущейВерсии");
    // Команда1НаСервере("СобиратьСтатистикуПоПоказателям");
    Команда1НаСервере("Константа1");
    Команда1НаСервере("Константа2");
    Команда1НаСервере("Константа3");
    Команда1НаСервере("Константа4");
    Команда1НаСервере("Константа5");
    КонецПроцедуры

    &НаСервере
    Процедура ЭлементПриИзмененииНаСервере(Путь)
    //Запись константы
    Константы[Путь].Установить(ЭтаФорма[Путь])
    КонецПроцедуры

    &НаКлиенте
    Процедура ЭлементПриИзменении(Элемент)
    ЭлементПриИзмененииНаСервере(Элемент.Имя);
    КонецПроцедуры
```
---
Спасибо за ответ, это все понятно, что через реквизит формы проблема решается ))